



Heuristic Code

```
// Halloween Eyes:
// Timer
unsigned long Timer;
#define TIMER_INTERVAL 1000
int RightFlag = false;
void setup() {
    // put your setup code here, to run once:
    DDRC |= 0x07; // A0, A1 and A2 are outputs.
    PORTC = 0x02; // Start off with middle LED dark
    Timer = millis();
} // End of Setup

// put your main code here, to run repeatedly:
void loop()
{
    if (millis() - Timer >= TIMER_INTERVAL)
    {
        if (PORTC & 0x01) // Right LED dark
        {
            // Turn on Middle and set RightFlag.
            PORTC = 0x02; // Middle LED dark
            RightFlag = true; // Got there from right
        }
        else if (PORTC & 0x04) // Left dark
        {
            // Turn on Middle and clear RightFlag.
            PORTC = 0x02; // Middle LED dark
            RightFlag = false; // Got there from right
        }
        else // assume middle LED is dark or PORTC = 0x02.
        {
            if (RightFlag) // if previously on right.
                PORTC = 0x04; // Left dark
            else
                PORTC = 0x01; // Right dark
        } // if for state of led's

        Timer += TIMER_INTERVAL;
    } // end of timer if.
} // end of loop
```

State Machine Approach

```
unsigned long Timer;
#define TIMER_INTERVAL 1000

// System for state transitions and bit pattern.
enum EyeStates { Left, MiddleLeft, MiddleRight, Right };
EyeStates eyeState = MiddleLeft;

// State Transition function,
// returns value for PORTC.
int NextState()
{
    int ReturnValue = 1; // default to Right dark
    switch (eyeState)
    {
        case Left: // Currently have left dark
            eyeState = MiddleRight; // middle state headed to right.
            ReturnValue = 0x02; // middle LED is dark
            break;
        case MiddleLeft: // Currently middle is dark
            eyeState = Left; // Move to left
            ReturnValue = 0x04; // setting left dark
            break;
        case MiddleRight: // Currently middle is dark,
            eyeState = Right; // Move to right
            ReturnValue = 0x01; // Setting right dark
            break;
        case Right: // Currently right is dark
            eyeState = MiddleLeft; // Middle state headed to left.
            ReturnValue = 0x02; // Setting middle dark
            break;
    } // end of switch
    return ReturnValue;
} // end of NextState

void setup() {
    // put your setup code here, to run once:
    DDRC |= 0x07; // Set A2, A1 and A0 to OUTPUT.
    PORTC = 0x02;
    Timer = millis();
} // end of setup

// put your main code here, to run repeatedly:
void loop()
{
    // Check if Interval has passed.
    if (millis() - Timer >= TIMER_INTERVAL)
    {
        // Move to next state
        PORTC = NextState(); // and set bit pattern
        Timer += TIMER_INTERVAL; // Update timer
    } // end of Timer loop.
} // end of loop.
```