

This writeup is meant to demonstrate the difference of integers versus unsigned integers in the C programming language. The following is some simple code that prints out a signed number and an unsigned number every second. The numbers are then incremented for the next printout.

```
// Two test variables,
// started close to their wrap point.
unsigned int usInt = 65528, RefusInt = 65528;
int Int = 32760, RefInt = 32760;
// Timer
unsigned long Timer;
#define PRINT_INTERVAL 1000
// Code run at start up.
void setup()
{
    // put your setup code here, to run once:
    Serial.begin(38400);
    Timer = millis();
} // End of setup

// Code that is run repeatedly
void loop()
{
    // Check if one second has passed
    if (millis() - Timer >= PRINT_INTERVAL)
    {
        // Send out the two variable values.
        Serial.print("Signed int = ");
        Serial.print(Int);
        Serial.print(" - ");
        Serial.print(RefInt);
        Serial.print(" = ");
        Serial.println(Int - RefInt);
        Serial.print("Unsigned ");
        Serial.print(usInt);
        Serial.print(" - ");
        Serial.print(RefusInt);
        Serial.print(" = ");
        Serial.println(usInt - RefusInt);

        // Update variables.
        Int++; usInt++;

        // Update timer to next point.
        Timer += PRINT_INTERVAL;
    } // End of Timer
} // End of loop.
```

```
Signed int = 32765 - 32760 = 5
Unsigned 65533 - 65528 = 5
Signed int = 32766 - 32760 = 6
Unsigned 65534 - 65528 = 6
Signed int = 32767 - 32760 = 7
Unsigned 65535 - 65528 = 7 Signed Wrap
Signed int = -32768 - 32760 = 8
Unsigned 0 - 65528 = 8 Unsigned Wrap
Signed int = -32767 - 32760 = 9
Unsigned 1 - 65528 = 9
Signed int = -32766 - 32760 = 10
Unsigned 2 - 65528 = 10
Signed int = -32765 - 32760 = 11
Unsigned 3 - 65528 = 11
Signed int = -32764 - 32760 = 12
```

Autoscroll Newline 38400 baud

In this screen shot of the print out, we can see that when a signed int reaches 32767, on the next increment it wraps around to -32768. Thus the range for a signed integer number is -32768 (-2^{15}) to 32767 ($2^{15}-1$). On the other hand an unsigned integer wraps around from 65535 to zero, and thus has a range of 0 to 65535 ($2^{16}-1$).

Also demonstrated in this example is the effect of mathematical operation on signed and unsigned numbers. Note in each case when we take the difference between the new value and the reference value (the starting value for each variable) the difference is unaffected by the wrap point and we have a difference that matches with the number of increments performed on each. The effect of wrapping on the difference will be important in the way we set up a time reference.